


Mode Mixer




Allows you to alter Display lists
AND
Generate ERROR free BASIC code

Margie Bliss

Jerry Kuit

BATTLE STATIONS



Challenge the computer or your friends!

Xlent
Software

by Johnny Masuda

P.O. Box 5228
Springfield, VA 22150
(703) 644-8881

Copyright (c) 1983

\$19⁹⁵

Table of Contents

| | Page |
|-------------------------------|------|
| Getting Started | 1 |
| Acknowledgements | 1 |
| Mode Mixer 1 | 2 |
| Overview | 2 |
| Using Mode Mixer | 2 |
| Using the Menu | 4 |
| Background Information | 4 |
| A Detailed Example | 4 |
| Creating Your Screen | 5 |
| User's Screen | 6 |
| Program Listing | 6 |
| Pokes 87, 88 and 89 | 6 |
| Get Printed Copy | 7 |
| Save to Disk | 9 |
| End Program | 9 |
| Using the Generated Program | 9 |
| Error Messages | 10 |
| Jump Instruction Error | 12 |
| Programming Notes | 13 |
| Quick Reference Sheet | 14 |
| Display List Generator | |
| (Mode Mixer 2) | 15 |
| Overview | 15 |
| Operation of Program | 15 |
| Creating a Custom Display | 17 |
| Using the Custom Display List | 19 |
| Notes on READDISP.LST | 21 |
| Error Messages | 21 |
| Battle Stations | 22 |

GETTING STARTED

Before running Mode Mixer 1 or 2, Make sure you have a formatted disk with DOS handy to save your work to.

1. Turn off all your equipment and remove any cartridges from your computer.
2. Turn on your T.V. and disk drive. If you have a printer and are going to run Mode Mixer 1, turn on the printer and interface.
3. When the busy light goes out on the disk drive, insert the Mode Mixer program disk and turn on your computer. (If you have more than one drive, the program disk must go in drive 1.) The Main Menu will automatically boot.
4. Type in the number of the program you want to run. After a second or two, the program will boot in and run and you're ready to start.

Acknowledgements

Mode Mixer 1 and Battle Stations were produced using copyrighted software products of Monarch Data Systems, Cohituate, MA 01778.

Mode Mixer 2 was compiled using DATASOFT'S BASIC COMPILER for the ATARI.

ATARI is a registered trademark of Atari, Inc.

Many thanks to Jennifer Brabson for the cover artwork and Mark Whiteside for the cover layout.

Documentation for Mode Mixer was printed using MegaFont.

We welcome any written comments or suggestions that will help us provide you with the software you want.

MODE MIXER: A Display List Utility
copyright (c) 1982 by Margie Bliss

OVERVIEW

MODE MIXER is a user friendly program designed to write the BASIC code needed to mix several graphics modes on one screen. No knowledge of the display list structure is needed to use this graphics program. A sample screen showing the relative sizes of each graphics mode is available through the menu. To create your custom screen, you simply enter the number of lines of each graphics mode you want, using the keyboard. MODE MIXER tells you how many lines of each mode you may enter, and when your screen is full, you see your custom screen displayed. Each mode on your screen is shown in alternating colors. You can quickly create one screen after another until you have one that satisfies you. The BASIC code to re-create this screen in your own program can then be listed to your television screen, to your printer, or to your own disk. The BASIC pokes necessary to correctly write or plot to your new screen are also provided. MODE MIXER takes the work out of creating a mixed mode screen by making all the calculations for you. Mixed mode screens of any combination can be quickly made by anyone.

USING MODE MIXER

THE MAIN MENU

The main menu appears after the opening display. The menu contains the following options:

SAMPLE SCREEN
CREATE YOUR SCREEN
USER'S SCREEN
PROGRAM LISTING
POKES 87, 88, AND 89
GET PRINTED COPY
SAVE TO DISK
END PROGRAM

The options are briefly described below:

OPTION ONE SAMPLE SCREEN

The sample screen is used for demonstration purposes. It displays each graphics mode available to you, in the size it will appear on your screen.

OPTION TWO CREATE YOUR OWN SCREEN

This option allows you to create your custom screen by entering each graphics mode desired on your screen, followed by the number of lines you wish to have of each mode. BASIC graphics modes zero through eleven may be used.

OPTION THREE USER'S SCREEN

This option displays the screen you have created, showing each graphics mode in alternating bands of color.

OPTION FOUR PROGRAM LISTING

Selecting this option lists the BASIC code MODE MIXER has written for you on your TV screen. This code can then be used to re-create your custom screen in your own program.

OPTION FIVE POKES 87, 88, AND 89

This option lists the pokes you will need to use in your program to correctly write or plot to your custom screen.

OPTION SIX GET PRINTED COPY

Use this option if you have a printer to get a printed copy of the graphics modes you selected, the program listing, and the pokes.

OPTION SEVEN SAVE TO DISK

This option allows you to save the listed version of the BASIC code written by MODE MIXER, on your own disk, for use in your own programs.

OPTION EIGHT END PROGRAM

Select this option when you have completed your screen. Insert a BASIC cartridge and load up DOS to create your own screens using the BASIC code saved either to your disk or printer.

USING THE MENU

To use the menu, press the OPTION key until the dark background color is underneath the option you wish to use. Then press the START key. For example, pressing the OPTION key once places the darker color underneath the option **SAMPLE SCREEN**. If you then press the START key, the sample screen will appear.

MODE MIXER is designed to be a completely self-prompting program. At this point, you may simply select the option **CREATE YOUR SCREEN** and begin. Just respond to the questions as they appear. If you wish to exit the routine before you have completely filled your screen type a 99. MODE MIXER will let you know if you make any errors. Detailed explanations of each menu option follows.

BACKGROUND INFORMATION

ATARI sets the information it needs to put graphics modes on the screen from the display list. ATARI automatically generates a display list for each graphics mode specified in a BASIC program. To use two or more graphics modes simultaneously on one screen, you must modify the display list, to tell ATARI which modes you wish to appear on the screen. The program listing MODE MIXER supplies you with is the set of BASIC commands needed to modify the machine generated display list to your own combination of modes. Pokes 87, 88, and 89 provide the computer with the information needed to write or plot to the new screen.

A DETAILED EXAMPLE

The following example will illustrate in detail how to use MODE MIXER to create a screen, and how to use the program listing and pokes supplied to you by MODE MIXER in your own program.

A demo screen is included on the MODE MIXER disk which is the following example. If you wish to see the

screen described in the text follow the directions to create the screen or select **END OF PROGRAM** to return to the main menu and then select **MM1 DEMO**.

This example will create a screen for a game. This screen consists of a line at the top for the name of the game, a large portion of the screen for the playfield, and some lines at the bottom for the score. Looking at the **SAMPLE SCREEN** option in **MODE MIXER**, mode two will be used for the name of the game, mode seven for the playfield, and mode one for the score. Press the **START** key to return to the menu, select the option **CREATE YOUR SCREEN**.

CREATING YOUR SCREEN

You will see a largely blank screen, with the prompt "ENTER FIRST GRAPHICS MODE DESIRED". Enter the number 2 from the keyboard and press the **RETURN** key. **MODE MIXER** will then tell you how many lines of mode 2 you may enter, followed by the question "HOW MANY LINES OF MODE 2 DO YOU WANT?". Enter a 1 and press **RETURN**. **MODE MIXER** will then ask you to "ENTER NEXT GRAPHICS MODE DESIRED". Enter a 7 and press **RETURN**. **MODE MIXER** will now tell you how many lines of mode 7 you may enter and ask you how many lines you want. Enter 76 and press **RETURN**. Again, you will be asked to enter the next mode. Enter a 1 and press **RETURN**. **MODE MIXER** will tell you how many lines of mode 1 you may enter. Enter a 3 and press **RETURN**. You will be informed that your screen is full, and will be given the choice of starting over or continuing on. Press the **OPTION** key to continue. You have now completed the first step, and a new screen will appear. This screen displays your selections. For the above example, the screen will appear as follows:

```
* YOU HAVE SELECTED *
  1 LINE OF GRAPHICS MODE 2
 76 LINES OF GRAPHICS MODE 7
  3 LINES OF GRAPHICS MODE 1
```

```
* TO SEE YOUR SCREEN PRESS OPTION *
* TO START OVER PRESS START *
```

When you see your selections listed, if you aren't satisfied with them, you may press the START key. MODE MIXER will return to the beginning of **CREATE YOUR SCREEN**. If you are satisfied with your selections, press the OPTION key. You then will be told to press START to return to the menu after viewing your screen. After a brief delay, you will see the screen you have created.

USER'S SCREEN

The screen you have created will appear while a tune plays. You will see three colors on the screen. The top color, a medium shade is the upper border of the screen. Your actual screen begins with the light shade. The first light colored section is the first graphics mode you selected--mode 2 in the example we are using. The next section of the screen, a dark color, is the second mode you selected--mode 7 in our example. The lowest section on the screen is always the bottom border. After viewing your screen, press START and you will be returned to the menu. You may select the option **USER'S SCREEN** if you wish to see the screen again. You may select this option as many times as you wish, however, every time you select the option **CREATE YOUR SCREEN**, you will lose any previously created screens.

PROGRAM LISTING

Select this option after you have created a screen you would like to include in your own BASIC program. The BASIC code written by MODE MIXER will be displayed. This is the code which may be saved to your own disk. In Figure 1, lines 10000 through 10050 show the program written by MODE MIXER for the example program.

POKE 87, 88, AND 89

Poke 87 is used to tell ATARI which graphics mode you are going to use. Poke 88 and 89 are screen memory locations which tell ATARI where to find the data to

put on the screen. You will need to use each of these pokes before writing or plotting to each mode in your custom screen. When you are writing to modes zero, one, or two, you must position the cursor before making your print statement. POSITION 0,0 will place the cursor in the upper left corner of the graphics area. The cursor must be positioned each time you change between these three graphics modes, as illustrated in Figure 2. Copy the pokes exactly as printed. NOTE * If you have a printer, you will not need to copy the pokes.

GET PRINTED COPY

Select this option if you have a printer. Figure 1 shows the printed copy as it appears for the example we are using. You may stop the printing routine at any time by pressing the START key.

```
      * YOU HAVE SELECTED: *  
      1 LINE OF GRAPHICS MODE 2  
      76 LINES OF GRAPHICS MODE 7  
      3 LINES OF GRAPHICS MODE 1  
  
10000 GRAPHICS 7+16  
10010 LBYTE=PEEK(560):HBYTE=PEEK(561):SCREEN=LBYTE+HBYTE*256+4  
10020 POKE SCREEN-1,64+7  
10030 FOR K=1 TO 76:POKE SCREEN+K+1,13:NEXT K  
10040 FOR K=1 TO 3:POKE SCREEN+K+77,6:NEXT K  
10050 POKE SCREEN+81,65:POKE SCREEN+82,LBYTE:POKE SCREEN+83,HBYTE  
  
      * GRAPHICS 2 SECTION *  
      POKE 87,2   POKE 88,96   POKE 89,144  
      * GRAPHICS 7 SECTION *  
      POKE 87,7   POKE 88,116  POKE 89,144  
      * GRAPHICS 1 SECTION *  
      POKE 87,1   POKE 88,84   POKE 89,156
```

Figure 1

```

10 REM #####
20 REM ***
30 REM ***   SAMPLE PROGRAM   ***
40 REM ***   USING MODE MIXER 1   ***
50 REM ***   GENERATED PROGRAM   ***
60 REM ***
70 REM #####
80 REM
90 REM **** MODIFIED DISPLAY LIST PROVIDED BY MODE MIXER ****
95 REM
100 TOP=PEEK(106):IF TOP>192 THEN POKE 106,192
10000 GRAPHICS 7+16:REM ****  USE THE GRAPHICS MODE REQUIRING THE MOST RAM ****
10004 REM
10005 REM **** FIND START OF THE DISPLAY LIST ****
10006 REM
10010 LBYTE=PEEK(560):HBYTE=PEEK(561):SCREEN=LBYTE+HBYTE*256+4
10020 POKE SCREEN-1,64+7:REM ****  1 LINE OF MODE 2 ****
10030 FOR K=1 TO 76:POKE SCREEN+K+1,13:NEXT K:REM ****  76 LINES OF MODE 7 ****
10040 FOR K=1 TO 3:POKE SCREEN+K+77,6:NEXT K:REM ****  3 LINES OF MODE 1 ****
10044 REM
10045 REM **** END OF DISPLAY LIST. JUMP TO THE START OF THE DISPLAY LIST ****
10046 REM
10050 POKE SCREEN+81,65:POKE SCREEN+82,LBYTE:POKE SCREEN+83,HBYTE
10052 REM **** NOW TO WRITE TO THE NEW SCREEN WE USE THE POKES FROM MODE MIXER.
10053 REM POKE 87 TELLS ATARI WHICH MODE TO USE. POKES 88 AND 89 TELL ATARI
10054 REM WHERE TO FIND THE DATA TO DISPLAY ON THE SCREEN ****
10055 REM
10056 SETCOLOR 4,8,8:SETCOLOR 0,8,2
10057 REM
10058 REM **** USE MODE 2 TO PRINT GAME NAME ****
10059 REM
10060 POKE 87,2:POKE 88,96:POKE 89,144:POSITION 3,0:PRINT #6;" GREAT GAME"
10061 REM
10062 REM **** NOTE:THE POSITION STATEMENT MUST BE USED TO PLACE THE CURSOR
10064 REM IN THE UPPER LEFT CORNER OF EACH SECTION OF A PRINT MODE SCREEN ****
10065 REM
10068 REM **** NOW USE MODE 7 AND PLOT AN OUTLINE OF THE PLAYFIELD ****
10069 REM
10070 POKE 87,7:POKE 88,116:POKE 89,144
10080 COLOR 1:PLOT 0,0:DRAWTO 159,0:DRAWTO 159,75:DRAWTO 0,75:DRAWTO 0,0
10085 REM
10086 REM **** NOW USE MODE 1 AND PRINT THE SCORE DATA FIELD ****
10087 REM
10090 POKE 87,1:POKE 88,84:POKE 89,156
20000 POSITION 0,0:PRINT #6;"PLAYER 1":PRINT #6;"PLAYER 2":PRINT #6;"HIGH SCORE"
21000 GOTO 21000

```

Figure 2

SAVE TO DISK

Select this option when you have created a screen you wish to use in your own BASIC program. You will need to insert your own disk and will be asked to name your file and enter your disk drive number. Any filename complying with the rules specified in the DOS Reference Manual may be used. You may include a three character extender with the filename. When you select this option, the space for the filename is colored a dark shade, to represent the length of an eight character filename plus an extender. It is not necessary to use the entire block. The BASIC code shown in the **PROGRAM LISTING** option will be saved with your filename and LISTed to the disk and can then be ENTERed as a part of your own BASIC program.

END PROGRAM

Select this option after you have created a screen you like, have saved the program to disk, have copied the pokes, or had them printed to your printer, or at any time your wish to quit. This option returns you the main menu.

USING THE MODE MIXER GENERATED PROGRAM IN YOUR OWN PROGRAM

If you have created a screen and saved it to disk you are now ready to recreate your custom screen in your own BASIC program. Insert a BASIC cartridge and replace the MODE MIXER disk with the disk on which your program was saved, type ENTER "D:FILENAME and LIST the program. The BASIC code from MODE MIXER has been given high line numbers to allow you to merge it with another program, if desired. Following the sample program shown in Figure 2, type in the pokes for each graphics mode before writing or plotting to that graphics mode on your custom screen. Figure 2 contains the BASIC code from MODE MIXER and includes remarks to explain each line. Each section of the screen is written or plotted to, to demonstrate the use of the pokes, When plotting to your own screen, don't forget to use a color statement in your plotting modes or you won't be able to see your drawings.

ERROR MESSAGES

Since MODE MIXER allows the use of the keyboard, there are instances where an incorrect keystroke may generate an error message. Most of the error messages in MODE MIXER are found in the **CREATE YOUR SCREEN** portion of the program.

Error messages are accompanied by a distinct sound, so if you are not looking at the screen, you will still be aware of the error. Information messages are displayed with a different sound, and you should be able to quickly distinguish between the two sounds.

The error messages are designed to be self-explanatory, for instance, entering a 12 as the next mode desired will prompt the message * ENTER ONLY MODES 0-11 *.

After receiving an error message, you will be returned to where you were before the error occurred, or given a choice of starting over or continuing on from that point.

If you select **USER'S SCREEN**, **PROGRAM LISTING**, **POKES 87, 88, AND 89**, **GET PRINTED COPY**, or **SAVE TO DISK** before you have created a screen, you will see the message * YOU MUST CREATE A SCREEN FIRST *. You will be returned to the menu. Before any of these five options can be executed, you will have to create a screen.

If you select the option **GET PRINTED COPY** and your printer or interface are not turned on, you will be asked to turn them on, and you may then press OPTION to continue. If for any reason, you are unable to continue, you may press START and return to the menu.

When you are using the **SAVE TO DISK** option, if you enter a disk drive number other than 1-4 you will be asked to use only number 1-4. After a delay, you will be returned to the previous line. If you enter a disk drive number for a disk drive that you do not have, or you encounter another disk error, you will see the

message * DISK ERROR NNN HAS OCCURED * *PLEASE BEGIN AGAIN *. At this point you may look up the error in the DOS Reference Manual. There will be a delay during which you may copy the error number. You will then be returned to the beginning of the **SAVE TO DISK** routine.

Recovering from any of these errors should cause you no problem. However, if at any time, you feel hopelessly lost, hitting the SYSTEM RESET key will return you to the main menu, but you will lose any previously created screens.

There are two types of errors which require a little more explanation. The first error occurs in the **CREATE YOUR SCREEN** portion of MODE MIXER. It is possible to create a screen which does not fill the entire available screen area. ATARI fills the screen with 192 scan lines. Various graphics modes use different amounts of these scan lines to make each of the mode lines. Graphics mode two uses sixteen scan lines to make one mode line, while graphics mode eight uses only one scan line for each mode line. If you create a screen which does not fill all 192 scan lines, you will be told your screen is not completely full, and you will be informed of the number of additional lines you may enter. At this point, you may continue on and see your selections and then your screen, or you may begin again. You also have the option of seeing your selections, to remind yourself what you chose, and from there beginning again.

It is not necessary to fill the entire screen at all times, and you may simply choose to ignore the warning that your screen is not completely full.

If you choose to ignore the warning that your screen is not full, and continue on, please look carefully at the screen you have created when it appears. The bottom color on the screen is always the bottom border, and you will not be able to write or plot to this section. When you use your custom screen in your own BASIC program and set the background and border colors to the same color, your screen will appear full. However, you may not be able to plot or write

as far down on the screen as you had wished, if you have not used all the scan lines. Since scan lines are so narrow, in some instances it will make little noticeable difference if you have not used all the available scan lines. This warning is included simply to make you aware there are more scan lines available if you wish to use them.

JUMP INSTRUCTION ERROR

The second error which requires a little more explanation is the jump instruction error. This error occurs most often in creating high resolution screens using modes eight through eleven. Because of the amount of memory required for fine resolution, these modes need to contain a jump instruction in the display list to cross a 4K address boundary. MODE MIXER will automatically write the BASIC code necessary to jump the 4K address boundary. However, in order to be able to correctly plot and draw on the high resolution section of a mixed mode screen, you must carefully plan the modes used above any high resolution mode.

The various modes available to you require different amounts of memory. Mode three, for instance only requires ten bytes per mode line, while mode eight requires forty bytes. In creating a mixed mode screen with high resolution modes, the modes used above these modes on your screen must equal a multiple of forty. Again using mode three as an example, you would need to use four (or eight or twelve, etc.) lines of mode three, if you wanted to use mode three above mode eight. You may consult the chart below to determine how many bytes each mode requires:

| Graphics Mode | Bytes Per Line |
|---------------|----------------|
| 0 | 40 |
| 1 | 20 |
| 2 | 20 |
| 3 | 10 |
| 4 | 10 |
| 5 | 20 |
| 6 | 20 |
| 7-11 | 40 |

You may use any combination of modes, as long as it results in a multiple of forty. Some examples could be:

- 1 line of mode 1 and 1 line of mode 2
- 2 lines of mode 2 and 1 line of mode 0
- 2 lines of mode 3 and 1 line of mode 1

You may use any number of lines of the forty byte mode lines. Modes 1-6 are the modes you need to consider carefully in planning a screen which includes a high resolution mode.

PROGRAMMING NOTES

USING MODES 9-11

In order to use modes 9-11 in your BASIC program, you will need to have the GTIA chip in your computer. If you do not know if you have this chip in your computer, you may enter this two line program.

```
10 GRAPHICS 10
20 GOTO 20
```

If the screen turns black, and remains black, you have the GTIA chip and may use modes 9-11 in your BASIC programs. If the screen changes to blue, you have the CTIA chip. You may still create screens in MODE MIXER using these modes, however, you will not be able to use them in your BASIC programs, unless you have the GTIA chip.

Please note that while Mode Mixer will generate the BASIC code to create a screen mixing CTIA and GTIA modes, machine language programming is generally needed to successfully use these screens. Advanced programming techniques such as display list interrupts are also required and it is considered beyond the scope of Mode Mixer to include this code. For more information on mixing CTIA and GTIA graphics, consult the articles on this subject which have appeared in computer magazines.

QUICK REFERENCE SHEET

1. Load **MODE MIXER** into your computer memory as explained in **Getting Started**. Don't forget to have a formatted disk for saving your work before you start.
2. Select option *** SAMPLE SCREEN *** and decide which modes you wish to use. Press **START** to return to the menu.
3. Select option *** CREATE YOUR SCREEN *** and enter the graphics modes and the number of lines you want of each mode, in the order you want them to appear on your screen.
4. If you are not satisfied with the appearance of your screen, select the option *** CREATE YOUR SCREEN *** again and make as many screens as you want until you have one you like.
5. If you are satisfied with the appearance of your screen, select the option *** PROGRAM LISTING *** to see the BASIC code **MODE MIXER** has written for you. This is the code which you may save to disk.
6. If you do not have a printer, select the option *** POKES 87, 88, AND 89 ***, and copy the pokes. If you have a printer, select the option *** GET PRINTED COPY ***. The pokes will be included in your printed copy.
7. Select the option *** SAVE TO DISK *** and replace the **MODE MIXER** diskette with your own DOS formatted diskette. Save the BASIC code to this diskette.
8. Select the option *** END PROGRAM ***. Insert your BASIC cartridge and load in your new program using the command **ENTER "D:FILENAME"**
9. When writing to your new screen, use the pokes from the *** POKES 87, 88, AND 89 *** option before writing to each graphics mode you selected.

DISPLAY LIST GENERATOR
(Mode Mixer 2)
REV 0.0 03/01/83
copyright (c) 1983 by JERRY KWIT

OVERVIEW

DISPLAY LIST GENERATOR creates custom display lists for later use in your own programs. You can preview your custom display before you save it to either tape or disk, and The DISPLAY LIST GENERATOR supports all twelve BASIC modes and the five ANTIC modes. All modes can be mixed on the screen by following the rules for inserting mode lines. The custom display lists are created from a table which prompts you for the GRAPHICS MODE you wish to use, and the number of lines of the GRAPHICS MODE. The program will edit all choices of GRAPHICS MODES and the number of lines selected, insuring a compatible selection of GRAPHICS MODES. The display can also be recreated if it doesn't meet your expectations.

OPERATION OF PROGRAM

The first display screen contains a table of BASIC modes and ANTIC modes in the first half, and the main rule for adding mode lines to your custom display in the second half. The first column of the table contains the BASIC mode numbers, and the ANTIC mode numbers. The ANTIC mode numbers do not correspond to the actual ANTIC mode numbers, they are numbered from twelve to sixteen for purposes of this program. Use these numbers for your displays. The second column tells you whether it is a text or graphics mode. The third column tells you the number of mode lines contained in the BASIC or ANTIC mode. The fourth column contains the number of scan lines in a line of the particular mode. Screens must contain 192 television scan lines. So for instance GRAPHICS 2 has 16 scan lines per mode line, and contains 12 mode lines ($16 \times 12 = 192$ scan lines). The fifth column indicates the number of bytes of screen memory needed for a mode line in each GRAPHICS mode or ANTIC mode.

The BASE GRAPHICS MODE is the mode you wish to use that uses the most memory. You can press <OPTION> at this point to check the RAM requirements for each BASIC GRAPHICS MODE. When prompted for BASE GRAPHICS MODE choose the BASIC MODE that you are going to use that needs the most memory and enter its number. ONLY BASIC GRAPHICS MODES ARE LEGAL FOR YOUR BASE GRAPHICS MODE, NOT ANTIC MODES.

Now the main rule for creating custom display lists. This will probably be confusing at first, but with a little practice you will be creating custom displays in no time.

"THE TOTAL NUMBER OF BYTES IN THE MODE LINES TO BE INSERTED INTO YOUR BASE GRAPHICS MODE, MUST BE A WHOLE MULTIPLE OF THE NUMBER OF BYTES IN A LINE OF YOUR BASE GRAPHICS MODE."

EXAMPLE: If you select GRAPHICS 7 as your base mode, and you want to insert some GRAPHICS 2 lines somewhere in the display, you must use at least 2 lines of GRAPHICS 2. 2 lines of GR.2 times 20 bytes per line = 40 bytes; GRAPHICS 7 contains 40 bytes per line, the GRAPHICS 2 lines contain 40 bytes so they are a whole multiple. $1 \times 40 = 40$.

If you tried to enter 3 lines of GRAPHICS 2 you would get an error message, "BYTES MUST BE A MULTIPLE OF MODE 7". 3 lines times 20 bytes per line of GR. 2 = 60 bytes. 60 bytes is not a whole multiple of 40 bytes. 40, 80, 120, 160, 200 are whole multiples of 40. This rule applies to all insertions of lines into your base mode.

CREATING A CUSTOM DISPLAY

Lets go through the process step by step and create a custom display.

1. LOAD THE PROGRAM FOLLOWING THE PREVIOUS DIRECTIONS.
2. STUDY THE TABLE AND READ THE DIRECTIONS.
3. HIT THE OPTION BUTTON FOR THE MEMORY TABLE.
4. FOR THIS EXAMPLE WE WILL BE USING GRAPHICS 7 AS OUR BASE MODE. AS YOU CAN SEE IT CONTAINS 4198 BYTES OF MEMORY.
5. HIT START. YOU ARE BEING PROMPTED FOR YOUR BASE MODE. ENTER 7 AND HIT RETURN. THE ASTERISK INDICATES YOUR BASE MODE.
6. AT THE TOP OF THE SCREEN WE WANT SOME GRAPHICS 2 LINES SO ENTER 2 IN RESPONSE TO THE GRAPHICS MODE PROMPT.
7. WE WANT 2 LINES OF GRAPHICS 2. ENTER 2. YOU'LL NOTICE THE "SCANS" COUNTER KEEPING TRACK OF YOUR TOTAL SCAN LINES. WE HAVE 32 SO FAR, WE NEED 192 FOR A COMPLETE DISPLAY.
8. ENTER 0 FOR GRAPHICS MODE.
9. ENTER 15 FOR LINES. 152 SCAN LINES SO FAR.
10. ENTER 7 FOR GRAPHICS MODE.
11. ENTER 20 FOR LINES. 192 SCAN LINES. A COMPLETE SCREEN.
12. HIT START TO PREVIEW DISPLAY.
13. IF YOU ENTERED THE INFORMATION CORRECTLY YOU SHOULD BE LOOKING AT YOUR FIRST CUSTOM DISPLAY. THE DISPLAY YOU SEE IS JUST RANDOM VALUES IN MEMORY.
14. HIT OPTION TO RETURN TO PROGRAM.
15. ENTER "R" AND YOU GET ANOTHER CHANCE TO TRY IT AGAIN.

```

3 REM *****
4 REM *      READDISP.LST      *
5 REM *      COPYRIGHT (C) 1983  *
6 REM *      BY JERRY KMIT      *
7 REM *****
10 GRAPHICS 8+16:REM USE YOUR BASE MODE
15 DIM RESV$(220),ANS$(20)
20 OPEN #3,4,0,"D:TESTLIST.LIS"
30 INPUT #3,RESV$:CLOSE #3
31 TLN=LEN(RESV$)
32 FOR I=1 TO 20:IF RESV$(I,I)="- " THEN NEXT I
34 RESV$=RESV$(I,TLN)
35 ADDR=ADR(RESV$):A2DR=ADDR:LN=LEN(RESV$)
50 LADDR=PEEK(560)
60 HADDR=PEEK(561)
70 POKE ADDR+(LN-2),LADDR:POKE ADDR+(LN-1),HADDR
80 POKE ADDR+4,PEEK(88):POKE ADDR+5,PEEK(89)
84 DLIST=PEEK(560)+256*PEEK(561):POKE 559,0:FOR I=DLIST TO DLIST+(LN-1):X=X+1
86 POKE I,ASC(RESV$(X,X)):NEXT I:POKE 559,34
95 REM *****
100 REM *****
110 POKE 87,8
112 TP5CRN=PEEK(88)+PEEK(89)*256
117 COLOR 1
120 PLOT 50,50:DRAMTO 75,99:PLOT 150,4:DRAMTO 75,99:PLOT 150,4:DRAMTO 225,99
125 PLOT 50,50:DRAMTO 225,99
130 POKE 87,1
150 TP5CRN=TP5CRN+4001
160 POKE 88,TP5CRN-(INT(TP5CRN/256)*256)
170 POKE 89,INT(TP5CRN/256)
180 POSITION 0,0:PRINT #6;"HI THERE, AND":POSITION 1,2:PRINT #6;"CONGRATULATIONS"
190 POKE 87,0
200 TP5CRN=TP5CRN+80
210 POKE 88,TP5CRN-(INT(TP5CRN/256)*256)
220 POKE 89,INT(TP5CRN/256)
230 POKE 752,1:POSITION 0,1:PRINT "          A CUSTOM DISPLAY.":INPUT ANS$
240 POKE 87,6
250 TP5CRN=TP5CRN+280
260 POKE 88,TP5CRN-(INT(TP5CRN/256)*256)
270 POKE 89,INT(TP5CRN/256)
280 COLOR 1:PLOT 4,1:DRAMTO 139,1
290 GOTO 290

```

Figure 3

Practice with the program for awhile and you'll find it becomes easy to create custom displays. The next time through try the save option.

USING THE CUSTOM DISPLAY LIST

To use your custom display after it has been saved to tape or disk, you read it into your own program with the code in Figure 3.

Change line 20 to OPEN #3,4,0,"C:" to load the Display List from the Cassette Recorder.

This short program is in LISTed format on the program disk as READDISP.LST. Lines 10 through 90 are the lines you need to read your custom display list in your own programs. Lines 110 through 290 are used to illustrate an example of writing to the screen using a custom display list. Lines 110 to 290 vary depending on the particular modes you are using. We will use this code to write to the screen, but first we will create the custom display list.

1. Load the DISPLAY LIST GENERATOR per the previous directions.
2. After you hit START enter 8 as your BASE MODE.
3. Enter 8 to the first GRAPHICS MODE prompt.
4. Enter 100 for the number of lines.
5. Enter 1 for the next GRAPHICS MODE.
6. Enter 4 for number of lines.
7. Enter 0 for GRAPHICS MODE.
8. Enter 7 for number of lines.
9. Enter 6 for GRAPHICS MODE.
10. Enter 2 for number of lines.
11. Hit START to preview display.

12. Hit OPTION to return to program.

13. READDISP.LST should be on the same disk as our Display List.

14. Save the DISPLAY LIST as TESTLIST. A file extender is affixed to the dataset name as '.LI5'. The program is set up to save Display Lists to Disk Drive 1.

15. Insert your BASIC cartridge and load READDISP.BAS.

16. Type RUN.

Leave the display on the screen while you read this next part. Look at lines 110 through 290 of READDISP.LST and notice the POKE statements at lines 110, 130, 190, and 240. These POKEs tell the Operating System what GRAPHICS MODE we want to use to write to the screen. GRAPHICS 8, OR 1, OR 0, OR 6. Always POKE location 87 with the GRAPHICS MODE you want to use BEFORE printing or plotting.

Now, since we have different mode lines on the screen that contain a different number of bytes in each line (remember column 5 of the Main table in the DISPLAY LIST GENERATOR), we have to tell the Operating System where the top of screen memory is for EACH GRAPHICS MODE.

For instance, look at line 112 of READDISP.LST. TP5CRN is assigned the location of the top left corner of the screen from locations 88 (low byte) and 89 (high byte). This corresponds to our GRAPHICS 8 section at the top of the screen. Look at line 150. 4001 is added to TP5CRN. 4000 is the number of bytes of GRAPHICS 8 we are using. 100 lines times 40 bytes per line = 4000 bytes. If we take this number and add 1 to it (4001), and then add this to TP5CRN, the location in TP5CRN now corresponds to the beginning of our GRAPHICS 1 section; "HI THERE". When we POKE this new TP5CRN back into locations 88 and 89, (lines 160 and 170), the Operating System thinks that this is now the top of screen memory and we can print "HI THERE" at position 0,0.

Notice lines 200 and 250. Do the same thing for each GRAPHICS MODE. Add the number of bytes in the previous mode to TP5CRN and POKE this new address into locations 88 and 89. The process for poking locations 88 and 89 is a standard procedure for splitting a number into a two byte address.

When creating your own displays, write down the GRAPHICS MODES, the number of lines in each mode, and the number of bytes in each line, so it will be easy to figure your TP5CRN location.

NOTES ON READDISP.LST

The GRAPHICS MODE in line 10 is the BASE MODE you used when you created your display list.

The OPEN statement in line 20 is the name of your custom display list.

LOCATION 87 POKES FOR ANTIC MODES

POKE Location 87 with BASIC MODE 0 for ANTIC MODE'S 12, 13, AND 14.

POKE Location 87 with BASIC MODE 7 for ANTIC MODE 15.

POKE Location 87 with BASIC MODE 8 for ANTIC 16.

If you are going to use a large amount of ANTIC MODE 16, your BASE MODE should be 8, 9, 10, or 11.

ERROR MESSAGES

If the number of lines you enter are not a multiple of bytes of the BASE MODE, error message 'BYTES MUST BE MULTIPLE OF MODE' will appear.

If your entry goes over 192 scan lines you will see error message 'TOO MANY SCAN LINES'. You can reenter after either of these messages.

BATTLE STATIONS
copyright (c) 1983
by Johnny Masuda

Challenge the computer or your friends to our version of this old pencil and paper favorite.

BOOTING BATTLE STATIONS

Following the instructions in Getting Started, boot up the Mode Mixer disk and select Battle Stations from the main menu.

The game will boot and you will see the screen in Figure 4.

Hit the Select button to choose the 1 or 2 player game.

Press Start to begin the game.

DEPLOY YOUR FLEET

You will now be asked to locate your fleet of 5 ships on the playing grid. Ships can be placed either horizontally or vertically, but may not overlap. Select "H" or "V".

Now the prompt will ask for the location to place one end of the ship. For horizontal ships, this is the left end, for vertical ships, this is the top. Enter the row number and column letter and hit return. Continue until all 5 ships are placed.

2 Player game: Ask your opponent to cover his or her eyes while you locate your fleet. No peeking between your fingers now...




Figure 4

BOMBS AWAY!!!!

Now you're ready to play. Take turns with your opponent entering your target coordinates. If you get a hit, a message will flash on the bottom of the screen telling you what was hit.

The fleet at the top of the screen keeps track of the damage you've inflicted on your opponent (or your opponent on you).

With Mode Mixer 1 & 2, you too can write games that are this attractive with little trouble. I hope that this will inspire you write your own applications. If you do write a good program, contact us at XLENT Software. We are always looking for good programs.



Other Programs From XLENT Software

MegaFont is the most enjoyable and useful printer utility ever. **MegaFont** is the first package that makes your printer fully compatible with your computer. It allows you to list ALL control and INVERSE characters to your printer in any of 9 fonts provided or create your own with any font editor. You can also dump Graphics 8 screens to the printer in 3 different sizes. Written by Richard Rognlie & Randy Dellinger for use with Prowriter, NEC or Epson w/Grafrax printers.

48K disk.....only \$19.95

MegaFont Picture Disk 1 provides you with 11 new pictures to dump with **MegaFont**. Artistic work by Jennifer Brabson & Mark Whiteside.

Disk.....\$10.00

C.A.M.P. (Computer Assisted Math Program) by Johnny Masuda is the BEST math drill program written for the ATARI. It is easily used by elementary students and has multiple skill levels to make it challenging to older children. Use your keyboard or joystick for input.

32K disk or tape...only \$19.95

If you have a buss-decoding modem from MicroPeripheral Corp., we have an X-modem protocol terminal program and a 64 column terminal program for it. Both programs are included for only \$29.95.

Please include \$2.00 for shipping and handling. Add \$1.50 for C.O.D.

**XLENT Software
P.O. Box 5228
Springfield, VA 22150
(703) 644-8881**

Necessary Equipment

ATARI computer with 48K RAM
ATARI compatible disk drive

MODE MIXER 1 by Margie Bliss

A user friendly program designed to write the BASIC code needed to mix several graphics modes on one screen. No knowledge of the display list structure is needed to use this graphics program. You can quickly create different custom screens using graphics modes 0 thru 11 until you have one you like. Then save the BASIC code that creates this screen to your disk or get a printed listing. Mode Mixer 1 takes the work out of creating a mixed mode screen by making the calculations for you.

MODE MIXER 2 by Jerry Kwit

For those more familiar with display lists, this program creates custom display list code for later use in your own programs. Preview your custom display before saving it to tape or disk. Supports all twelve BASIC modes and the five ANTIC modes. All modes can be mixed on the screen by following the rules for inserting mode lines. The program edits all your choices of graphics modes and the number of lines selected to insure compatible selections.

BATTLE STATIONS by Johnny Masuda

Just like that old game played with pencil and paper. Challenge your friends or the computer. Strategically locate your fleet, then BOMBS AWAY!!!!

XLENT Software Replacement Policy

We will gladly send you one backup copy for \$5.00. This fee covers the cost of the new media, handling and return postage within the United States and Canada. Please include the serial number of your program with your request. After the first backup, please return the original copy with the \$5.00 fee.